
Appendix A. TOPAS input file showing the CaF2 structural model and the nontronite supercell 

model used for supercell mode calibration. The Rietveld quantification formula is reversed and 

used to refine the interlayer cation occupancy of the nontronite supercell model. This is used 

to determine the Bulong nontronite unit cell mass and volume with known weight fractions of 

standard mixtures. The bold text highlights the novel calibration coding developed for this 

paper. 

'------------------------------------------------------------------------------- 

'CaF2 generated from best CaF2 synchrotron pattern fitting.  

str  

 prm cmCaF2 = Get(cell_mass); 

 prm cvCaF2 = Get(cell_volume); 

 LVol_FWHM_CS_G_L( 1, 0, 0.89, 0,@, 1000 min =500; max =10000;,@, 1000 min =500; max =10000;) 

 e0_from_Strain( 0,@, 0.05 min =0; max =0.05;,@, 0.05 min =0; max =0.05;) 

 r_bragg 1.39482605 

 phase_name "Calcium Fluorite" 

 MVW(312.300, 163.213765`_0.000194236569, 100.000`_0.000) 

 scale !scCaF2 0.00554598749 

 space_group Fm-3m 

 Phase_LAC_1_on_cm( 110.76187`_0.00013) 

 Phase_Density_g_on_cm3( 3.17734`_0.00000) 

 Cubic(!afluorite  5.46494`_0.00000 min =5.45; max =5.49;) 

 site Ca1 num_posns  4 x =0; : 0.00000 y =0; : 0.00000 z =0; : 0.00000 occ Ca+2 1 beq 0.5888 min 0.1 max 1 

 site F1 num_posns  8 x =1/4; : 0.25000 y =1/4; : 0.25000 z =1/4; : 0.25000 occ F-1 1 beq 0.7937 min 0.1 max 1 

'------------------------------------------------------------------------------- 

'turbostratically disordered smectite d(001) approx. 15 A 

'structure of TOT layer (cis-vacant) by: Tsipurski et al., Clay Minerals 19(1984), 177-193 (modified) 

'model for turbostratic disorder by: Ufer et al., Z. Kristallogr. 219(2004), 519-527 

str  

 prm interlayer 0.39816`_0.00631 

 'local scnon 0.00041`_0.00000 min 0.0000001 max 1 'scale factor 

 prm mshkl 0.01999`_0.03602_LIMIT_MIN_5e-005 min 0.00005 max 0.02 'strain of non-basal reflection 

 prm ms00l 6.73911`_0.32386 min 0.001 max 10 'strain of basal reflection 

 prm cshkl 15.63019`_0.40663 min 10 max 1000 'size of non-basal reflections 

 prm cs00l 9.20520`_0.17236 min 1 max 20 'size of basal reflections 

 

 'occupancies of cations 

 prm !pMg 0.065 min 0 max 1 prm !pNi 0.065 min 0 max 1 prm !pFe 0.6 min 0 max 1 prm !pAl 0.2375 min 0 max 

1 'octahedral position, Mg + Fe + Al +Ni < 1. 

 prm !ptrans 1 min 0 max 1 'mixing parameter for cis- and trans-vacancy; 0 => trans-vacant. 

 prm !altet 0.08375 min 0 max 1 prm sitet =1-altet; 0.91625 'tetrahedral site: Al + Si = 1. 

 prm pNa = interlayer 0.717;:0.31488`_0.00098 min 0 prm pMgI = interlayer 0.287;:0.12604`_0.00039 min 0 

prm pH2O = pNa+pMgI;: 0.44092`_0.00105 min 0 'interlayer H2O 

 'atom absolute position in c direction to avoid a stretching/shortening of the TOT layer by varying c0 

 prm !zT 2.7135  

 prm !zO11 1.0955  

 prm !zO12 1.0553  

 prm !zO2 3.3668  

 

 prm !layer 9 'layer: supercell factor for elongation in c direction 

 prm cnon 15.61679`_0.00516 min 14 max 15.8 'subcell c 

 

 lor_fwhm = If(And(H==0,K==0), 0.1 Rad Lam/(Cos(Th) cs00l), 0.1 Rad Lam/(Cos(Th) cshkl)); 'thinner in 00l 

direction  

 lor_fwhm = If(And(H==0,K==0), ms00l Tan(Th), Sqrt((mshkl Tan(Th))^2 + L^2/(Get(c)^2 L^2 + Get(c)^4 

((H/Get(a))^2+(K/Get(b))^2))) ); 'aniso strain and additional l-dependent broadening to avoid "ripples"  

 scale_pks = If(And(H==0,K==0),If(Mod(L,layer),0,layer),1); 'scaling of classes (00l and hkl) and removal of 

redundant 00l reflections 

 'scale =scnon;  

 scale = cmCaF2 cvCaF2 

scCaF2/Get(cell_mass)/Get(cell_volume)/0.100948577134299*0.899051422865701; :8.1020183e-006_4.48e-007 

 

r_bragg  2.35613318 

 phase_name "Bulong supercell" 

 MVW( 999.686`_0.654, 6620.83728`_2.49863214, 98.741`_0.070) 

 'corrected_weight_percent  42.9984672_0.601779145 

 space_group 5 

 Phase_LAC_1_on_cm( 6.48560`_0.00594) 

 Phase_Density_g_on_cm3( 0.25073`_0.00019) 

 a anon  5.25575`_0.00063_LIMIT_MIN_5.25 min 5.25 max 5.35 

 b !bnon  9.09592434 min 9.1 max 9.2  

 c =layer*cnon; 'supercell c 

 be benon  99.81479`_0.04551 

 

 'Na+0.716Mg2+0.287[Si4+7.33Al3+0.67][Al3+0.95Fe3+2.40Ni2+0.26Co2+0.01Mg2+0.26Cr3+0.09]O20(OH)4  

'Bulong Nontronite 

'octahedral trans 

 site Al1 num_posns 2  x =0;:0.00000  y =0;:0.00000  z =0;:0.00000 occ Al+3 =ptrans*pAl; beq 0.5 

site Mg1 num_posns 2  x =0;:0.00000  y =0;:0.00000  z =0;:0.00000 occ Mg+2 =ptrans*pMg; beq 0.5 

 site Fe1 num_posns 2  x =0;:0.00000  y =0;:0.00000  z =0;:0.00000 occ Fe+3 =ptrans*pFe; beq 0.5 

 site Ni1 num_posns 2  x =0;:0.00000  y =0;:0.00000  z =0;:0.00000 occ Ni+2 =ptrans*pNi; beq 0.5 

'octahedral cis 

 site Al2 num_posns 2  x =0;:0.00000  y 0.6540  z =0;:0.00000 occ Al+3 =(1-ptrans)*pAl; beq 0.5 

site Mg2 num_posns 2  x =0;:0.00000  y 0.6540  z =0;:0.00000 occ Mg+2 =(1-ptrans)*pMg; beq 0.5 

 site Fe2 num_posns 2  x =0;:0.00000  y 0.6540  z =0;:0.00000 occ Fe+3 =(1-ptrans)*pFe; beq 0.5 

 site Ni2 num_posns 2  x =0;:0.00000  y 0.6540  z =0;:0.00000 occ Ni+2 =(1-ptrans)*pNi; beq 0.5 

 site Al3 num_posns 2  x =0;:0.00000  y 0.3210  z =0;:0.00000 occ Al+3 =pAl; beq 0.5 

 site Mg3 num_posns 2  x =0;:0.00000  y 0.3210  z =0;:0.00000 occ Mg+2 =pMg; beq 0.5 

 site Fe3 num_posns 2  x =0;:0.00000  y 0.3210  z =0;:0.00000 occ Fe+3 =pFe; beq 0.5 

 site Ni3 num_posns 2  x =0;:0.00000  y 0.3210  z =0;:0.00000 occ Ni+2 =pNi; beq 0.5 

'tetrahedral 

 site Si1 num_posns 4  x 0.4320  y =1/3;:0.33333  z =zT/(layer*cnon); occ Si+4 =sitet; beq 0.4 

 site Al4 num_posns 4  x 0.4320  y =1/3;:0.33333  z =zT/(layer*cnon); occ Al+3 =altet; beq 0.4 

 site Si2 num_posns 4  x 0.4320  y =2/3;:0.66667  z =zT/(layer*cnon); occ Si+4 =sitet; beq 0.4 

 site Al5 num_posns 4  x 0.4320  y =2/3;:0.66667  z =zT/(layer*cnon); occ Al+3 =altet; beq 0.4 

 

 site O7 num_posns 4  x 0.1730  y 0.7250  z =zO2/(layer*cnon); occ O-2 1 beq 0.8 

 site O8 num_posns 4  x 0.1700  y 0.2680  z =zO2/(layer*cnon); occ O-2 1 beq 0.8 

 site O9 num_posns 4  x 0.4170  y 0.6560  z =zO11/(layer*cnon); occ O-2 1 beq 0.8 

 site O10 num_posns 4  x 0.3430  y 0.3470  z =zO11/(layer*cnon); occ O-2 1 beq 0.8 

 site O11 num_posns 4  x 0.3340  y -0.0240  z =zO12/(layer*cnon); occ O-2 1 beq 0.8 

 site O12 num_posns 4  x 0.4890  y 0.4960  z =zO2/(layer*cnon); occ O-2 1 beq 0.8 

'interlayer 

 site Mg num_posns  4  x 0.50000  y 0.25000  z 0.05556 occ Mg+2 =pMgI; beq 0.8 

 site Na num_posns  4  x 0.50000  y 0.25000  z 0.05556 occ Na+1 =pNa; beq 0.8 

 site O1 num_posns  4  x 0.35860  y 0.43735  z 0.06560 occ O-2 =pH2O; beq 2.5 

 site O2 num_posns  4  x 0.64140  y 0.06265  z 0.04551 occ O-2 =pH2O; beq 2.5 

 site O3 num_posns  4  x 0.07980  y 0.25000  z 0.04551 occ O-2 =pH2O; beq 2.5 

 site O4 num_posns  4  x 0.92020  y 0.25000  z 0.06560 occ O-2 =pH2O; beq 2.5 

 site O5 num_posns  4  x 0.35860  y 0.06265  z 0.06560 occ O-2 =pH2O; beq 2.5 

 site O6 num_posns  4  x 0.64140  y 0.43735  z 0.04551 occ O-2 =pH2O; beq 2.5 

'rigid body of the interlayer complex  

prm !dMgO 2.41 

rigid 

point_for_site Mg 

point_for_site Na 

point_for_site O1 uz = dMgO; 

point_for_site O2 uz =-dMgO; 

point_for_site O3 ux = dMgO; 

point_for_site O4 ux =-dMgO; 

point_for_site O5 uy = dMgO; 

point_for_site O6 uy =-dMgO; 

 

rotate 45 qx 1 operate_on_points "O*" 

rotate =180/Pi*ArcSin(1/Sqrt(3)); qy 1 operate_on_points "O*" 

rotate 180 qz 1 operate_on_points "O*" 

translate tx =0.5*anon+0.5*cnon*Cos(benon Deg); ty =0.5*bnon; tz =0.5*cnon*Sin(benon Deg); operate_on_points "O* 

Mg Na" 

 




